N63-21345

NASA TECHNICAL NOTE

NASA TN D-1902

NASA TN D-1902

ANALYSIS OF THE DYNAMIC TESTS OF THE STRETCH YO-YO DE-SPIN SYSTEM

by William R. Mentzer
Goddard Space Flight Center
Greenbelt, Maryland

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION - WASHINGTON, D. C. - SEPTEMBER 1963

TECHNICAL NOTE D-1902

ANALYSIS OF THE DYNAMIC TESTS OF THE STRETCH YO-YO DE-SPIN SYSTEM

By William R. Mentzer

Goddard Space Flight Center
Greenbelt, Maryland

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

ANALYSIS OF THE DYNAMIC TESTS OF THE STRETCH YO-YO DE-SPIN SYSTEM

by
William R. Mentzer
Goddard Space Flight Center

SUMMARY

Results of the stretch yo-yo feasibility and flight qualification tests are presented. These tests were conducted to prove the concept that the stretch yo-yo is a more accurate de-spin device than the rigid yo-yo, and to verify the analytical development of the stretch yo-yo properties. Variations in the design parameters and their effects on the final spin rate of the payload are noted in the analysis of the test results. The variables include initial spin rate, moment of inertia, and spring properties. A computer solution of the test payload equations of motion is included for comparison with the experimental results to confirm the mathematical analysis of the stretch yo-yo system. As a result of the successful flight qualification tests a stretch yo-yo was flown on Ariel I (1962 o1) in April 1962.

CONTENTS

Summary i

INTRODUCTION. 1

OBJECTIVES OF THE TESTS. 2

TEST APPARATUS 3

TEST PROCEDURE 3

DISCUSSION OF TEST RESULTS. 6

References 9

Appendix A—Computer Solution of Phase 1 Equations of Motion . . 11

ANALYSIS OF THE DYNAMIC TESTS OF THE STRETCH YO-YO DE-SPIN SYSTEM

by
William R. Mentzer
Goddard Space Flight Center

INTRODUCTION

The stretch yo-yo is a de-spin device that has the ability to compensate for errors in the initial spin rate and the moment of inertia of a payload. It is composed of an end mass, a helical spring, and, if necessary, a length of wire (Figures 1 and 2). The spring elongates under a load, giving a variable yo-yo length. The square of the yo-yo length varies directly as the initial spin rate, and the ratio of final spin to initial spin varies inversely as the square of the yo-yo length. This produces an essentially constant final spin rate. The concept of the stretch yo-yo was suggested by H. J. Cornille



Figure 1—Stretch yo-yo consisting of wire, spring, and end mass. (The scale is in inches.)



Figure 2—Stretch yo-yo consisting of a spring and end mass. (The scale is in inches.)

(Reference 1). The analytical theory and design criterion were developed by Dr. J. V. Fedor (References 2 and 3).

Following dynamic analysis and development of design equations for the stretch yo-yo, the decision was made to fabricate and test stretch yo-yos on the Explorer XII (1961 υ1) and Ariel I (1962 o1) type payloads. Three series of tests were conducted beginning with feasibility tests on the two payloads and concluding with flight qualification tests for the Ariel I payload. As a result of these tests, a stretch yo-yo was flown on the successful Ariel I spacecraft in April 1962.

In the record of the experimental results, the effects of the several variables involved in stretch yo-yo design can be readily noted. These variables include optimum spring constant, preload, initial spin rate, spin moment of inertia, and material strength.

OBJECTIVES OF THE TESTS

The stretch yo-yo tests had several objectives which overlapped the three series of tests. The first series of tests was intended to demonstrate the feasibility of the stretch yo-yo de-spin system in compensating for errors in initial spin rate. This series also was used to examine the characteristics of a helical spring operating as a yo-yo.

The second test series was planned as the flight qualification for the Ariel I stretch yo-yo made of National Standard Company's NS355 high-strength stainless steel. The yo-yos were to be tested for sensitivity to changes in spin moment of inertia, and for performance at overspin conditions when the yo-yos would be operating near the yield point of the material.

Because of an unexpected delay, the steel shipment did not arrive in time for the scheduled tests. Therefore, it was necessary to fabricate springs from conventional music wire. This change in spring material necessitated a change in the test series. The objects of the revised test series were to observe the effect of changes in moment of inertia, as originally planned, and to study the results of subjecting the spring to stresses greater than the yield point of the material.

When the NS355 steel was received and the Ariel I springs had been fabricated, the third, and last, test series was conducted. These tests qualified the stretch yo-yo as Ariel I flight hardware. The yo-yos were tested at overspin conditions to see if they could withstand the loading.

TEST APPARATUS

Tests on the stretch yo-yo were performed in the vacuum facilities at Langley Research Center, Hampton, Virginia. The vacuum tanks were spheres 41 and 60 feet in diameter. The test structures were the portions of the outer shells of the payloads (Explorer XII and Ariel I) on which the yo-yos were positioned, and inertia plates for simulating the inertias of the complete payloads. A dc motor, sealed for vacuum operation, with an electromagnetic drive unit and an electromagnetic coupling and brake comprised the spin table drive system. The yo-yo firing signal was fed from an external manually operated circuit to the payload through a set of slip rings in the driven shaft of the spin table. Payload angular velocity was measured in the following manner. A disk with 32 equally spaced, radially protruding studs was mounted on the driven shaft; then, as the shaft turned, the studs generated pulses in an adjacent magnetic pickup. The output signal from this circuit was fed into a recording oscillograph. In order to obtain a complete time record of the operation, the declutching signal, firing signal, and yo-yo release signal also were fed into the oscillograph.

The spin table and the mounting platform in the 60 foot vacuum chamber at Langley Research Center are shown in Figure 3. Figure 4 shows the Ariel I payload mounted on the spin table. The yo-yos can be seen in the picture. The details of the spin table with its drive mechanism and spin rate measuring device are shown in Figure 5.

TEST PROCEDURE

The original intent for the tests was to study only the ability of the stretch yo-yo to compensate for errors in the initial spin rate of the payload and errors in payload moment of inertia. But, as a result of yo-yo fabrication problems, the experimenters were forced to consider the effects on the final spin rate of preload in the yo-yos, of deviation from the optimum spring constant, and of spring loading in the region of the elastic limit of the material. Because of the increased complexity of the program, it was necessary to be careful in pairing the springs for the test. The spring scale and the preload were determined for each spring to be tested. Heat treating and flexing were performed on the springs in an attempt to decrease the preloads. One spring from each group fabricated was statically loaded until permanent deformation occurred. This was done in order to obtain a strength limit to



Figure 3—The Ariel I payload mounted on the spin table in the 60 foot vacuum chamber.



Figure 4—Closeup of the Ariel I payload showing the yo-yos in place. Weights at the base of the structure are for simulating the entire payload moment of inertia.

compare with the expected maximum yo-yo tension in the upcoming tests. With regard to the two factors affecting the tests, spring constant and preload, a deviation from the optimum spring constant was the variation most critical to the test results. In view of this fact, the springs with essentially equal spring constants were paired for testing. The pairs were then tested in the order of smallest to largest preload.

After the yo-yo springs were paired, the average values of spring scale and preload for a set were used to compute the end mass (Reference 2). The yo-yos were assembled so that both springs in each test were of the same length.

The tests were conducted at an absolute pressure of 10 mm Hg to minimize the atmospheric drag effects on the yo-yos. Payload spin rate was determined from the recording oscillograph records (Figure 6).



Figure 5—Details of the spin table.



Figure 6—A copy of the data tape used for the Explorer XII type stretch yo-yo, test IV, November 14, 1961.

The oscillograph output consisted of a plot of pulses from the magnetic contact on a known time scale. Also included on the record were the declutching signal, the yo-yo firing signal, and the yo-yo release signal.

The spin table was constructed so that one pulse from the magnetic contact took 1/32 of a revolution of the payload and that the time scale of the recorder chart τ was 1/100 of a second per division. The spin rate at the midpoint of the time interval between any two pulses was found from the relation

$$W = 1.875 \frac{N}{t} ,$$

where

W = spin rate in rpm,

N = number of pulses,

t = total time between pulses in seconds.

DISCUSSION OF TEST RESULTS

The stretch yo-yo tests were considered successful from several standpoints. The concept of the stretch yo-yo as a more accurate de-spin device than the rigid yo-yo was proven, and the analytically developed design criteria were verified (Reference 1). The difficulties of operating with a non-optimum yo-yo system were evaluated. Finally, the yo-yo was qualified as flight hardware for Ariel I.

The results of the stretch yo-yo tests are recorded in Table 1. The tests are grouped by series and by payload and are numbered consecutively from the first successful test on each payload. System design parameters and actual test conditions have been tabulated in order that the effect on final spin rate of variations of test conditions from design values can be readily noted.

The Explorer XII type optimum spring scale tests clearly demonstrated the ability of the stretch yo-yo to compensate for errors in initial spin rate. In this series of tests, spin-up errors of ±20 percent were reduced to within ±1.5 percent of the design final spin rate. The fine performance was attributed to the fact that the yo-yo springs had the optimum spring constant and no preload. The yo-yos used in these tests were of the type shown in Figure 1.

The importance of the optimum spring constant in the proper functioning of the stretch yo-yo de-spin system was illustrated by the last three tests on the Explorer XII configuration. These tests were necessitated by the manufacturing of preloaded springs with incorrect spring constants. The tests were conducted at a higher spin level than tests II through IV because of apprehension concerning the functioning of a preloaded yo-yo. Spin limits were selected and calculations were made by using the measured values of preload and spring constant, which were nonoptimum for the spin rates used.

The percent error shown for these tests is large when referenced to the 30 rpm design final spin. If the results are referenced to test V, with a 26.8 rpm final spin rate, as design conditions, then the

Table 1

Stretch Yo-Yo Test Summary.

Date	Payload	Test	Design Parameters: Initial Spin Rate $\dot{\phi}_0$ (rpm)	Design Parameters: Final Spin Rate $\dot{\phi}_f$ (rpm)	Design Parameters: Spin Moment of Inertia I (slug-ft^2)	Design Parameters: Spring Constant (optimum) k (lb$_f$/ft)	Test Conditions: Spin Moment of Inertia I (slug-ft^2)	Test Conditions: Spring Constant (average) k (lb$_f$/ft)	Test Conditions: Preload (average) F_0 (lb$_f$)	Test Spin Data: Initial Spin Rate $\dot{\phi}_0$ (rpm)	Test Spin Data: Error in Initial Spin Rate (percent)	Test Spin Data: Final Spin Rate $\dot{\phi}_f$ (rpm)	Test Spin Data: Error in Final Spin Rate (percent)
November 13-16, 1961	Explorer XII	II	103.0	14.48	2.135	7.88	2.135±0.1	7.745	0.	101.81	-1.156	14.42	-0.415
		III	123.5					7.76	0.	120.75	+17.23	14.48	0.0
		IV	82.5					7.66	0.	82.50	-19.90	14.69	+1.45
		V	110.0	30.00				12.00	2.72	108.75	-2.028	26.79	-10.7
		VI	132.0					11.63	2.70	130.31	+17.40	27.50	-8.33
		VII	88.0					12.16	2.62*	88.13	-20.60	20.46	-31.80
	Ariel I	II	160.0	74.80	2.774	11.00	2.774±0.1	10.85	9.325	162.56	+1.60	73.56	-1.658
		III	128.0					11.29	8.855	129.38	-19.14	73.75	-1.404
January 22-24, 1962	Ariel I	IV	160.0	70.79	2.885	11.56	2.885±0.1	11.565	3.40	155.66	-2.712	71.20	+0.534
		V	184.0					11.53	3.35	184.42	+15.26	61.30†	-13.44
		VI	136.0					11.58	3.50	138.89	-13.19	72.81	+2.807
		VII	160.0					11.605	3.675	157.90	-1.313	71.66	+1.183
		VIII	160.0		3.462		3.462±0.1	11.51	3.15	160.26	+0.162	76.53	+8.06
		IX	128.0					11.64	3.20	131.25	-17.97	76.85	+8.512
		X	160.0		2.308		2.308±0.1	11.405	3.81	154.41	-3.494	66.96	-5.452
		XI	192.0					11.91	3.475	187.50	+17.19	57.25†	-19.16
		XII	128.0					11.625	3.75	129.31	-19.18	64.21	-9.335
February 7, 1962	Ariel I	XIII	160.0	73.90	2.885	12.48	2.885±0.1	11.93	1.90	160.71	+0.444	77.16	+4.411
		XIV	184.0					11.915	2.175	182.96	+14.35	74.57	+0.906
		XV	200.0					11.87	2.025	200.38	+25.54	72.58	-1.787

*One yo-yo had zero preload.
†Yielding occurred.

overspin error is + 2.7 percent. The underspin error, -23.6 percent, is still large but is attributed to the fact that the test was made with yo-yos of unequal preloads, differing by 2.6 lb.

Ariel I yo-yo tests II and III proved the feasibility of the one-half wrap stretch yo-yo system (Figure 1) and led to the use of the stretch yo-yo on this satellite. Tests IV and VII demonstrated the repeatability of the yo-yo results and, along with test VI, gave a picture of the system operating at the design moment of inertia.

The design final spin rate for these tests was computed from equations that included the effects of preload in the yo-yo springs. The value of the preload used in calculations was that of the springs used in test IV. Test results agreed closely with theoretically predicted final spin rates, the deviation resulting from the fact that the spring constants varied slightly from the optimum spring constant.

In tests V and XI the springs yielded because the load exceeded the yield point of the material. The yielding produced a spin rate that was much lower than theoretically predicted. This situation was permitted to occur since the springs being tested were of lower strength limits than flight hardware. The experimenters were afforded the opportunity to observe the behavior of the de-spin system when yielding occurred and the design equations no longer applied.

Tests VIII through XII verified the theoretical prediction that the stretch yo-yo would be relatively insensitive to variations in spin moment of inertia when compared with the behavior of the rigid yo-yo. Moment-of-inertia values of ± 20 percent of the design spin axis moment of inertia were used in the tests.

The stretch yo-yo tests were concluded with the Ariel I flight qualification tests, XIII through XV. In these tests, the NS355 high-strength stainless steel springs were used. The tests were conducted at design initial spin rate and then at + 15 and + 20 percent of design initial spin rate to determine whether the yo-yo could withstand the high loading at overspin conditions. Test XIII, with nominal initial spin rate, gave a final spin slightly higher than design final spin. In the overspin tests, the final spin rates were very close to design values. As a result of this test series, the stretch yo-yo of high-strength stainless steel was accepted as flight hardware for Ariel I which was launched successfully in April 1962.

An analytical method for determining the actual final spin rate of a satellite is given in Reference 2. The data necessary for performing the calculations include spring properties, design values of spin rate and inertia, and actual values of spin rate and inertia.

The equations of motion of the test payload were programmed for solution on an IBM 7090 digital computer as a verification of the test results. Correspondence between this solution and the test results proved the validity of these equations of motion. The equations were developed for phase 1 of the yo-yo operation, when the yo-yo is unwinding and is tangential to the payload. Spin table friction was included in the analysis. The development of the equations and the computer program is included as Appendix A.

A graph of the computer solution of phase 1 for one test is included with the plot of the entire despin test in Figure 7. The plot shows the angular displacement of the payload, in radians, versus time.

The test data points are the pulses from the magnetic contact that were recorded by the recording oscillograph. Thirty-two successive points define one revolution of the payload. The end of phase 1 is determined by the computer program, and the end of phase 2 is indicated on data tape by the yo-yo release signal. At the end of phase 1 the computed angular displacement for test XIII of February 7, 1962, exceeded the recorded value by 2.42 percent.

Phase 2, when the yo-yo moves from a tangential to a radial position, was not analyzed because of algebraic complexity and similarity to phase 1 analysis. It was felt that the close agreement between the computer solution of the phase 1 analysis and the test results, combined with the similarity of the methods of solving phase 1 and 2 equations on a computer, justified the omission of the phase 2 analysis from this report.



Figure 7—Angular displacement vs. time for Ariel I test XIII, February 7, 1962.

REFERENCES

1. Cornille, H. J., Jr., "A Method of Accurately Reducing the Spin Rate of a Rotating Spacecraft," NASA Technical Note D-1420, October 1962.

2. Fedor, J. V., "Analytical Theory of the Stretch Yo-Yo for De-Spin of Satellites," NASA Technical Note D-1676, April 1963.

3. Fedor, J. V., "Theory and Design Curves for a Yo-Yo De-Spin Mechanism for Satellites," NASA Technical Note D-708, August 1961.

Appendix A

Computer Solution of Phase 1 Equations of Motion

The theoretical development and the design criteria for the stretch yo-yo de-spin mechanism functioning on an orbiting spacecraft have been developed by J. V. Fedor.* The stretch yo-yo tests were performed to demonstrate the feasibility of the stretch yo-yo system and to verify the design parameters that had been developed. Since the observed final spin rates of the payloads corresponded to the theoretical predictions, the tests were considered successful.

The differential equations of motion of the test payload on the spin table were solved on an IBM 7090 computer for comparison with the experimental results to verify the analysis of the stretch yo-yo system. The equations were developed for phase 1 of the yo-yo operation when the yo-yo is unwinding and tangential to the payload (Figure A1). In the analysis, the effects of gravity are considered negligible because of the short operating time of the yo-yo, 0.50 to 0.75 seconds, and because of the existence of a component of the tension in the yo-yo spring which opposes the gravitational force. The validity of this assumption has been demonstrated in previous rigid yo-yo de-spin tests where test results agreed closely with theoretical calculations in which gravitational effects were neglected. Tests were conducted at a pressure of 10 mm Hg, which corresponded to an altitude of 100,000 feet and made the effect of atmospheric drag negligible.

The coordinate system for phase 1 analysis is shown in Figure A1. Because of the symmetry of the system only one yo-yo is shown.

The stretch yo-yo de-spin system behaves as a rigid yo-yo until the preload is overcome by the force in the yo-yo. During this phase of operation the Lagrangian L, which is the kinetic energy of the system, is†

$$L = \frac{1}{2} I \dot{\phi}^2 + \frac{1}{2} m \left(l^2 \dot{\theta}^2 + a^2 \dot{\phi}^2 \right) , \quad \text{(A1)}$$



Figure A1—Phase I coordinate system.

*Fedor, J. V., "Analytical Theory of the Stretch Yo-Yo for De-Spin of Satellites," NASA Technical Note D-1676, April 1963.
†Fedor, J. V., "Theory and Design Curves for a Yo-Yo De-Spin Mechanism for Satellites," NASA Technical Note D-708, August 1961.

where

I = spin moment of inertia less yo-yo masses,

$\dot{\phi}$ = spin rate of payload,

θ = generalized coordinate,

m = mass of both yo-yos,

a = radius of yo-yo fixture,

l = length of yo-yo unwound.

The length of wire unwound at any time in rigid yo-yo operation is

$$l = a(\theta - \phi) . \tag{A2}$$

The mass of the yo-yo system, including the spring mass, is given by an approximate equation that accounts for the distributed spring weight,

$$m = m_0 + \frac{\rho a}{3} (\theta - \phi) , \tag{A3}$$

in which $(\rho a/3)(\theta - \phi)$ is the mass of yo-yo spring unwound for both yo-yos. The quantity ρ is twice the mass density of one helical spring and m_0 is the total end mass of the yo-yos. The equations of motion in terms of the Lagrangian are

$$\frac{d}{dt}\left(\frac{\partial L}{\partial \dot{\phi}}\right) - \frac{\partial L}{\partial \phi} = Q_\phi , \tag{A4}$$

$$\frac{d}{dt}\left(\frac{\partial L}{\partial \dot{\theta}}\right) - \frac{\partial L}{\partial \theta} = Q_\theta . \tag{A5}$$

Evaluation of these expressions, in which the generalized forces Q_ϕ and Q_θ consist only of the friction torque in the spin table, $Q_\phi = -sI$ and $Q_\theta = 0$, yields

$$\frac{d}{dt}\left[\left(I + ma^2\right) \dot{\phi}\right] + ma\, l\dot{\theta}^2 = -sI , \tag{A6}$$

$$\frac{d}{dt}\left(m\, l^2\, \dot{\theta}\right) - ma\, l\dot{\theta}^2 = 0 , \tag{A7}$$

where s is the deceleration rate of the payload due to friction measured at $t = 0$.

In order to obtain a difference equation for ϕ as a function of time, integrate Equation A6:

$$\left[\left(I + ma^2\right) \dot{\phi}\right]_k^{k+1} = -\int_k^{k+1} \left(sI + ma\, l\dot{\theta}^2\right) dt .$$

Then

$$\left(I + m_{k+1}\, a^2\right) \dot{\phi}_{k+1} = \left(I + m_k\, a^2\right) \dot{\phi}_k - sI\,\Delta t - m_k\, a l_k\, \dot{\theta}_k^2\, \Delta t . \tag{A8}$$

Represent $\dot{\phi}_{k+1}$ by a forward difference, $\left(\phi_{k+2} - \phi_{k+1}\right)/\Delta t$, and then solve Equation A8 for ϕ_{k+2}:

$$\phi_{k+2} = \phi_{k+1} + \frac{\Delta t}{I + m_{k+1}\, a^2} \left[\left(I + m_k\, a^2\right) \dot{\phi}_k - sI\,\Delta t - m_k\, a l_k\, \dot{\theta}_k^2\, \Delta t\right] . \tag{A9}$$

Performing the indicated differentiation of the second equation of motion yields

$$m\ddot{\theta} + \dot{m}\dot{\theta} = \frac{1}{l}\left(ma\,\dot{\theta}^2 - 2m\,\dot{l}\dot{\theta}\right), \tag{A10}$$

and the left-hand side of this expression is an exact differential which can be expressed as

$$\frac{d}{dt}\,(m\dot{\theta}) = \frac{m\dot{\theta}}{l}\,(a\dot{\theta} - 2\dot{l}) \ . \tag{A11}$$

Integration of Equation A11, by the method of Equation A8, combined with the replacement of $\dot{\theta}_{k+1}$ by a forward difference, gives the relation for θ_{k+2}:

$$\theta_{k+2} = \theta_{k+1} + \dot{\theta}_k\,\Delta t\,\frac{m_k}{m_{k+1}}\left[1 + \left(2a\,\dot{\phi}_k - a\dot{\theta}_k\right)\frac{\Delta t}{l_k}\right] \ . \tag{A12}$$

The computer solution is started by calculating the values of ϕ and θ for $t = 0$. At $t = 0, l = 0$ and the right-hand side of Equation A12 becomes infinite; thus, an expression must be developed to evaluate θ_{k+2} at the time $t = 0$.

From the expanded form of the second equation of motion, Equation A10, an expression for $\dot{\theta}_0$ is determined. Substitution of initial conditions into Equation A10 yields

$$\dot{\theta}_0 = 2\dot{\phi}_0 \ . \tag{A13}$$

Application of L'Hospital's rule to the term $\left(2a\,\dot{\phi}_k - a\dot{\theta}_k\right)\Delta t/l_k$ from Equation A12 yields, at $t = 0$,

$$\left(2\ddot{\phi}_0 - \ddot{\theta}_0\right)\frac{\Delta t}{\dot{\phi}_0} \ .$$

From the differentiation in Equation A6 the general expression for the payload angular acceleration is

$$\ddot{\phi}_k = \frac{-\dot{m}_k\,a^2\,\dot{\phi}_k - m_k\,a\,l_k\,\dot{\theta}_k^2 - sI}{I + m_k\,a^2}, \tag{A14}$$

which can be evaluated at $t = 0$:

$$\ddot{\phi}_0 = \frac{-\frac{\rho a}{3}\,\dot{\phi}_0^2 - sI}{I + m_0\,a^2} \ . \tag{A15}$$

Upon differentiating in Equation A7 and solving for $\ddot{\theta}$, it is found that this expression is undefined at $t = 0$; and L'Hospital's rule must be applied again. The resulting expression for $\ddot{\theta}$ at $t = 0$ is

$$\ddot{\theta}_0 = \frac{-2\rho a\dot{\phi}_0^2}{9m_0} + \frac{4\ddot{\phi}_0}{3} . \qquad \text{(A16)}$$

A combination of $(2\ddot{\phi}_0 - \ddot{\theta}_0)\,\Delta t/\dot{\phi}_0$ with Equation A12 yields an expression for θ_2 in terms of the initial conditions:

$$\theta_2 = \theta_1 + \left[1 + \frac{\Delta t\left(2\ddot{\phi}_0 - \ddot{\theta}_0\right)}{\dot{\phi}_0}\right]\frac{m_0}{m_1}\,\dot{\theta}_0\,\Delta t , \qquad \text{(A17)}$$

where $\theta_1 = \dot{\theta}_0\,\Delta t + \theta_0$.

The total force in the yo-yo system during the time when it performs essentially as a rigid yo-yo is given by

$$F_k = m_k\left(a\ddot{\phi}_k + l_k\,\dot{\theta}_k^2\right) + \dot{m}_k a\dot{\phi}_k . \qquad \text{(A18)}$$

The rigid yo-yo operation is terminated when F becomes greater than $2F_0$, the total preload in the yo-yo springs.

After the tensile force in the yo-yo has exceeded the preload, the yo-yo functions as a stretch yo-yo. The Lagrangian for the stretch yo-yo in phase 1 of yo-yo operation is*

$$L = \frac{1}{2}I\dot{\phi}^2 + \frac{1}{2}m\left(l^2\dot{\theta}^2 + a^2\dot{\phi}^2 + \dot{\delta}^2 - 2a\dot{\phi}\dot{\delta}\right) - \left(k\delta^2 + 2F_0\,\delta\right) , \qquad \text{(A19)}$$

in which

δ = deflection of a spring,

k = spring constant,

F_0 = preload in one spring.

The length of the stretch yo-yo at any time t is a function of the deflection and the amount of spring unwound; thus

$$l = a(\theta - \phi) + \delta ; \qquad \text{(A20)}$$

whereas the mass remains dependent only on the length of wire unwound,

$$m = m_0 + \frac{\rho a}{3}(\theta - \phi) .$$

*Fedor, J. V., "Analytical Theory of the Stretch Yo-Yo for De-Spin of Satellites," NASA Technical Note D-1676, April 1963.

Writing the equations of motion explicitly from the Lagrangian results in:

$$\frac{d}{dt}\left[\left(I + ma^2\right)\dot{\phi} - ma\,\dot{\delta}\right] + ma\,l\dot{\theta}^2 = -sI\ , \tag{A21}$$

$$\frac{d}{dt}\left(ml^2\,\dot{\theta}\right) - ma\,l\dot{\theta}^2 = 0\ , \tag{A22}$$

$$\frac{d}{dt}\left(m\,\dot{\delta} - ma\,\dot{\phi}\right) - ml\dot{\theta}^2 + 2k\delta + 2F_0 = 0\ . \tag{A23}$$

To determine ϕ as a function of time for stretch yo-yo operation, integrate Equations A21 and A23 in the manner of the preceding work to get

$$\left(I + m_{k+1}a^2\right)\dot{\phi}_{k+1} - m_{k+1}\,a\dot{\delta}_{k+1} - \left(I + m_k\,a^2\right)\dot{\phi}_k + m_k\,a\dot{\delta}_k = -m_k\,a\,l_k\,\dot{\theta}_k^{\,2}\,\Delta t - sI\,\Delta t\ , \tag{A24}$$

$$m_{k+1}\,\dot{\delta}_{k+1} - m_{k+1}\,a\dot{\phi}_{k+1} - m_k\,\dot{\delta}_k + m_k\,a\dot{\phi}_k = m_k\,l_k\,\dot{\theta}_k^{\,2}\,\Delta t - 2k\delta_k\,\Delta t - 2F_0\,\Delta t\ . \tag{A25}$$

Solve Equation A25 for $m_{k+1}\,\dot{\delta}_{k+1}$ and substitute into Equation A24, simultaneously replacing $\dot{\phi}_{k+1}$ by a forward difference, $\left(\phi_{k+2} - \phi_{k+1}\right)/\Delta t$, to get an expression for the angular coordinate of the payload:

$$\phi_{k+2} = \phi_{k+1} + \frac{\Delta t}{I}\left(I\dot{\phi}_k - sI\,\Delta t - 2ak\delta_k\,\Delta t - 2aF_0\,\Delta t\right)\ . \tag{A26}$$

A treatment of Equation A22 by the same method of integration and substitution yields

$$\theta_{k+2} = \theta_{k+1} + \frac{\Delta t}{m_{k+1}\,l_{k+1}^{2}}\left(m_k\,a\,l_k\,\dot{\theta}_k^{\,2}\,\Delta t + m_k\,l_k^2\,\dot{\theta}_k\right)\ , \tag{A27}$$

where $\dot{\theta}_{k+1} = \left(\theta_{k+2} - \theta_{k+1}\right)/\Delta t$ as above.

To determine the difference equation for δ, differentiate in Equation A23 as indicated and then simplify to the form:

$$\frac{d}{dt}\left(m\dot{\delta}\right) = \dot{m}a\dot{\phi} + ma\ddot{\phi} + ml\dot{\theta}^2 - 2k\delta - 2F_0\ . \tag{A28}$$

Integrate this expression and substitute a backward difference for $\dot{\delta}$ to get

$$\delta_{k+1} = \delta_k + \frac{\Delta t}{m_{k+1}}\left[m_k\,\dot{\delta}_k + \left(\dot{m}_k\,a\dot{\phi}_k + m_k\,a\ddot{\phi}_k + m_k\,l_k\,\dot{\theta}_k^{\,2} - 2k\delta_k - 2F_0\right)\Delta t\right]. \tag{A29}$$

This relationship is dependent upon $\ddot{\phi}$, which is determined by differentiating in Equation A21 as indicated and substituting from Equation A28:

$$\ddot{\phi}_k = -\frac{1}{I}\left(sI + 2ka\delta_k + 2F_0 a\right) \quad . \tag{A30}$$

If there is no preload in the yo-yo springs, the device will function as a stretch yo-yo from the start of the de-spin operation. For this reason it is necessary to be able to evaluate the expressions for ϕ, θ, δ, and $\ddot{\phi}$ at $t = 0$. Differentiation in Equation A22 gives

$$m\ddot{\theta} + \dot{m}\dot{\theta} = \frac{1}{l}\left(ma\dot{\theta}^2 - 2m\dot{l}\dot{\theta}\right) \quad , \tag{A31}$$

and substitution of initial conditions yields

$$\dot{\theta}_0 = 2\dot{\phi}_0 \quad ,$$

which is also true for rigid yo-yo operation. At $t = 0$, Equation A30 reduces to

$$\ddot{\phi}_0 = -s \quad , \tag{A32}$$

which is compatible with the physics of the problem in that the payload is decelerating because of friction at $t = 0$. The deflection equation becomes

$$\delta_1 = \frac{\Delta t^2}{m_1}\left(\dot{m}_0 a\dot{\phi}_0 + m_0 a\ddot{\phi}_0\right) \quad , \tag{A33}$$

at $t = 0$. Equation A26 for the payload angular coordinate takes the form:

$$\phi_2 = \phi_1 + \frac{\Delta t}{I}\left(I\dot{\phi}_0 - sI\,\Delta t\right) \quad , \tag{A34}$$

for $t = 0$. Equation A27 for θ cannot be evaluated at $t = 0$; thus, another formulation of the problem must be considered. Rewrite Equation A31:

$$\frac{d}{dt}(m\dot{\theta}) = \frac{1}{l}\left(ma\dot{\theta}^2 - 2m\dot{l}\dot{\theta}\right) \quad .$$

Then integrate and make a forward difference substitution for $\dot{\theta}$:

$$\theta_{k+2} = \theta_{k+1} + \dot{\theta}_k\,\Delta t\,\frac{m_k}{m_{k+1}}\left[1 + \left(2a\dot{\phi}_k - a\dot{\theta}_k - 2\dot{\delta}_k\right)\frac{\Delta t}{l_k}\right] \quad . \tag{A35}$$

This equation cannot be directly evaluated at $t = 0$. Apply L'Hospital's rule and initial conditions to the quantity $(\Delta t/l_k)\ (2a\dot{\phi}_k - a\dot{\theta}_k - 2\delta_k)$ to get $(2\ddot{\phi}_0 - \ddot{\theta}_0)\Delta t/\dot{\phi}_0$, where $\ddot{\theta}_0$ is found from Equation **A31** by the same method used in deriving Equation **A16:**

$$\ddot{\theta}_0 = \frac{-2\rho a\dot{\phi}_0^2}{9m_0} + \frac{4\ddot{\phi}_0}{3} . \tag{A36}$$

And we have for Equation A35

$$\theta_2 = \theta_1 + \left[1 + \frac{\Delta t}{\dot{\phi}_0}\left(2\ddot{\phi}_0 - \ddot{\theta}_0\right)\right]\frac{m_0}{m_1}\dot{\theta}_0\,\Delta t , \tag{A37}$$

with $\ddot{\phi}_0$ from Equation A32 and $\ddot{\theta}_0$ from Equation A36.

Phase 1 of the yo-yo operation is terminated when the entire yo-yo is unwound from the payload. The computer terminates the program when the relation

$$\left[B - a\left(\theta_k - \phi_k\right)\right] = 0 \tag{A38}$$

is satisfied, where B is the length of the yo-yo and $a(\theta_k - \phi_k)$ is the length of yo-yo unwound at time $t = t_k$.

An analysis of phase 2 of the yo-yo has been omitted from this report. Since the equations are similar to those for phase 1* and the computer solution of phase 1 was sufficiently close to the test data, it was felt that solution of the phase 2 equations would not add sufficient information to the report to warrant the additional effort involved.

The computer program is outlined on the following pages in the form of a block diagram, Figure A2, and a listing of the program in Fortran for the IBM 7090, Table A1. The nomenclature for the program is as follows:

ϕ = PHI,	$\dot{\phi}$ = OMEGA,	$\ddot{\phi}$ = ALPHA,
θ = THETA,	$\dot{\theta}$ = ETA,	$\ddot{\theta}$ = BETA,
δ = DELTA,	$\dot{\delta}$ = ZETA,	
l = EL,		
m = EM,	$\dot{m}$ = DEEM,	
F = FORCE,		
t = TIME,		

and in the input data:

a = A,	l_{SPRING} = B,	$(I + m_0 a^2)\dot{\phi}_0$ = PI3,
$\dot{\phi}_0$ = OMEGA0,	Δt = DTIME,	sI = PI4,
m_0 = EM0,	I = PI1,	2k = PI5,
$2F_0$ = P0,	$\rho a/3$ = PI2,	$2F_0$ = PI6.

*Fedor, J. V., "Analytical Theory of the Stretch Yo-Yo for De-Spin of Satellites," NASA Technical Note D-1676, April 1963.



Figure A2—Flow chart for the solution of the phase 1 equations of motion of the stretch yo-yo for the IBM 7090 digital computer.

Table A1

Fortran Computer Program for Phase 1.

```
* W.R.MENTZER        MECHANICAL SYSTEMS BRANCH       G.S.F.C.
*       XEQ
*       CARDS COLUMN
*       LIST 8
C       STRETCH YO YO     PHASE 1
 1001 FORMAT(3XF6.4,2XF8.4,3XF8.6,3XF7.4,3XF7.4,3XF7.5,2XF6.3/4XF9.8,3XF
     C9.6,3XF10.8,3XF7.3,3XF6.3)
 1002 FORMAT(6X1HA,5X6HOMEGA0,6X3HEM0,9X2HP0,7X1HB,8X5HDTIME,6X3HPI1,6X3
     CHPI2,9X3HPI3,9X3HPI4,6X3HPI5,6X3HPI6/3XF6.4,2XF8.4,3XF8.6,3XF7.4,3
     CXF7.4,3XF7.5,2XF6.3,2XF9.8,3XF9.6,3XF10.8,3XF5.2,2XF6.3///)
 1003 FORMAT(1H1,20X51HPROGRAM FOR THE TEST OF THE STRETCH YO YO   PHASE
     C1 //5X4HTIME,4X10HSPIN RATIO,4X9HSPIN RATE,6X3HPHI,6X12HACCELERATI
     CON,5X5HTHETA,8X6HLENGTH,5X10HDEFLECTION,5X4HMASS,6X5HFORCE/5X5H(SE
     CC),17X9H(RAD/SEC),5X5H(RAD),5X13H(RAD/SEC/SEC),4X5H(RAD),9X4H(FT),
     C9X4H(FT),7X7H(SLUGS),5X4H(LB)///)
 1004 FORMAT(3XF7.4,2XF11.8,3XF10.6,3XF10.7,3XF10.7,3XF10.7,3XF10.6,2XF1
     C1.8,3XF9.8,2XF7.4/)
 1005 FORMAT(///6X31HEND OF OPERATION AS RIGID YO YO )
 1006 FORMAT(3XF7.4,2XF11.8,3XF10.6,3XF10.7,3XF10.7,3XF10.7,3XF10.6,2XF1
     C1.8,3XF9.8/)
 1007 FORMAT(///6X14HEND OF PHASE 1 )
 1008 FORMAT(6X12HNEGATIVE EL1)
      READ  INPUT  TAPE 2,1001,A,OMEGA0,EM0,P0,B,DTIME,PI1,PI2,PI3,PI4,P
     CI5,PI6
      WRITE OUTPUT TAPE 3,1002,A,OMEGA0,EM0,P0,B,DTIME,PI1,PI2,PI3,PI4,P
     CI5,PI6
      WRITE OUTPUT TAPE 3,1003
      J=1
    1 TIME=0.0
      FORCE1=0.0
      EM1=EM0
      EL1=0.0
      OMEGA1=OMEGA0
      PHI1=0.0
      PHI2=OMEGA1*DTIME
      ETA1= 2.0*OMEGA0
      THETA1=0.0
      THETA2=ETA1*DTIME
      X1=PI1+EM1*A**2
      IF (P0)      9,110,10
    9 CALL EXIT
  110 TIME=TIME+DTIME
      DELTA1=0.0
      ZETA1=0.0
      EM2=EM0+PI2*(THETA2-PHI2)
      EL2=A*(THETA2-PHI2)
      PHI3=PHI2+(DTIME/PI1)*(PI1*OMEGA1-PI4*DTIME)
      OMEGA2=(PHI3-PHI2)/DTIME
      SMALR2=OMEGA2/OMEGA0
      ALPHA1= -PI4/PI1
      DEEM1=(EM2-EM1)/DTIME
      DELTA2=(DTIME**2/EM2)*(DEEM1*A*OMEGA1+EM1*A*ALPHA1)
      ZETA2=(DELTA2-DELTA1)/DTIME
      BETA1=-(2.0*PI2*OMEGA1**2)/(3.0*EM0)+(4.0*ALPHA1)/3.0
      THETA3=THETA2+DTIME*ETA1*(EM1/EM2)*(1.0+(DTIME/OMEGA1)*(2.0*ALPHA1
     C-BETA1))
      ALPHA2=-(1.0/PI1)*(PI4+A*PI5*DELTA2+PI6*A)
      GO TO 50
```

```
   10 TIME=TIME+DTIME
      EM2=EM0+PI2*(THETA2-PHI2)
      EL2=A*(THETA2-PHI2)
      X2=PI1+EM2*A**2
      PHI3=PHI2+(DTIME/X2)*(X1*OMEGA1-PI4*DTIME-EM1*A*EL1*ETA1**2*DTIME)
      OMEGA2=(PHI3-PHI2)/DTIME
      SMALR2=OMEGA2/OMEGA0
      IF (EL1)       13,14,15
   13 WRITE OUTPUT TAPE 3,1008
      CALL EXIT
   14 ALPHA1=-(PI2*A**2*OMEGA1**2 +PI4)/X1
      BETA1=-(2.0*PI2*OMEGA1**2)/(3.0*EM0)+(4.0*ALPHA1)/3.0
      THETA3=THETA2+DTIME*ETA1*(EM1/EM2)*(1.0+(DTIME/OMEGA1)*(2.0*ALPHA1
     C -BETA1))
      GO TO 16
   15 THETA3=THETA2+DTIME*ETA1*(EM1/EM2)*(1.0+(2.0*A*OMEGA1-A*ETA1)*(DTI
     CME/EL1))
      GO TO 16
   16 ETA2=(THETA3-THETA2)/DTIME
      ALPHA2= -(PI2*A**2*OMEGA2*(ETA2-OMEGA2)+EM2*A*EL2*ETA2**2+PI4)/X2
      EM3=EM0+PI2*(THETA3-PHI3)
      DEEM2=(EM3-EM2)/DTIME
      FORCE2=EM2*(A*ALPHA2+EL2*ETA2**2)+A*DEEM2*OMEGA2
   20 IF(51 - J)     21,22,23
   21 CALL EXIT
   22 WRITE OUTPUT TAPE 3,1003
      WRITE OUTPUT TAPE 3,1004, TIME,SMALR2,OMEGA2,PHI2,ALPHA2,THETA2,EL
     C2,DELTA2,EM2,FORCE2
      J=1
      GO TO 25
   23 WRITE OUTPUT TAPE 3,1004, TIME,SMALR2,OMEGA2,PHI2,ALPHA2,THETA2,EL
     C2,DELTA2,EM2,FORCE2
      J=J+1
      GO TO 25
   25 IF(P0-FORCE2)  40,40,30
   30 PHI2=PHI3
      OMEGA1=OMEGA2
      ALPHA1=ALPHA2
      THETA2=THETA3
      ETA1=ETA2
      EL1=EL2
      EM1=EM2
      X1=X2
      GO TO 10
   40 WRITE OUTPUT TAPE 3,1005
      PHI1=PHI2
      PHI2=PHI3
      OMEGA1=OMEGA2
      ALPHA1=ALPHA2
      THETA1=THETA2
      THETA2=THETA3
      ETA1=ETA2
      EL1=EL2
      EM1=EM2
      DELTA1=0.0
      ZETA1=0.0
      WRITE OUTPUT TAPE 3,1003
      J=1
   41 TIME=TIME+DTIME
      EM2=EM0+PI2*(THETA2-PHI2)
      DEEM1=(EM2-EM1)/DTIME
```

```
      PHI3=PHI2+(DTIME/PI1)*(PI1*OMEGA1-PI4*DTIME-A*PI5*DELTA1*DTIME-PI6
     C*A*DTIME)
      OMEGA2=(PHI3-PHI2)/DTIME
      SMALR2=OMEGA2/OMEGA0
      DELTA2=DELTA1+(DTIME/EM2)*(EM1*ZETA1+(DEEM1*A*OMEGA1+EM1*A*ALPHA1+
     CEM1*EL1*ETA1**2-PI5*DELTA1-PI6)*DTIME)
      ZETA2=(DELTA2-DELTA1)/DTIME
      EL2=A*(THETA2-PHI2)+DELTA2
      ALPHA2= -(1.0/PI1)*(PI4+A*PI5*DELTA2+PI6*A)
      THETA3=THETA2+(DTIME/(EM2*EL2**2))*(EM1*A*EL1*ETA1**2*DTIME+EM1*EL
     C1**2*ETA1)
      ETA2=(THETA3-THETA2)/DTIME
   50 IF( 51 - J)      51,52,53
   51 CALL EXIT
   52 WRITE OUTPUT TAPE 3,1003
      WRITE OUTPUT TAPE 3,1006, TIME,SMALR2,OMEGA2,PHI2,ALPHA2,THETA2,EL
     C2,DELTA2,EM2
      J=1
      GO TO 60
   53 WRITE OUTPUT TAPE 3,1006, TIME,SMALR2,OMEGA2,PHI2,ALPHA2,THETA2,EL
     C2,DELTA2,EM2
      J=J+1
      GO TO 60
   60 IF(B-A*(THETA2-PHI2))    90,90,80
   80 PHI2=PHI3
      OMEGA1=OMEGA2
      ALPHA1=ALPHA2
      THETA2=THETA3
      ETA1=ETA2
      EL1=EL2
      EM1=EM2
      DELTA1=DELTA2
      ZETA1=ZETA2
      GO TO 41
   90 WRITE OUTPUT TAPE 3,1007
      CALL EXIT
      END
*     DATA
```

THE DATA CARDS FOR THIS PROGRAM USING THE DATA FROM TEST III OF FEB. 7 1962 ARE AS FOLLOWS

A	OMEGA0	EM0	P0	B	DTIME	PI1
.942	16.8300	.033850	3.8000	2.3650	.0002	2.885

PI2	PI3	PI4	PI5	PI6
.00336171	49.060077	.40181588	23.860	3.800

NASA-Langley, 1963 G-394